PRNC-78

PUERTO RICO NUCLEAR CENTER

FORTRAN PROGRAM FOR CALCULATION OF NEUTRON DIFFRACTION MAGNETIC INTENSITIES



OPERATED BY UNIVERSITY OF PUERTO RICO UNDER CONTRACT NO. AT (40-1)-1833 FOR U. S. ATOMIC ENERGY COMMISSION

PRNC Report

Fortran Program for Calculation of Neutron Diffraction Magnetic Intensities

M.I. Kay

and

Don T. Cromer

Introduction

This report presents a Fortran program for computing the intensities of neutron diffraction magnetic reflections. It has been written for the IBM 1620 computer located at the University of Puerto Rico, Mayaguez, Puerto Rico. Consequently, all of the input-output is in the form of READ and PRINT or PUNCH statements. These statements are duplicated with similar tape statements preceded by C* punched in cols. 1 and 2. Another user might want to remove the READ, PUNCH and PRINT statements and replace them with the tape statements after deleting the C*.

Another user might also want to use the major portion of this program as a subroutine and write a main program which would systematically generate indices and/or spin vectors.

The Intensity Function

The program computes $|F(\underline{H})|^2 = F(\underline{H}) \times F(\underline{H})^*$

$$F(\underline{H}) = \underline{\varepsilon}(\underline{\varepsilon}\cdot\Sigma_j p_j \underline{K}_j \exp(2\pi i \underline{H}\cdot\underline{r}_j)) - \Sigma_j p_j \underline{K}_j \exp(2\pi i \underline{H}\cdot\underline{r}_j), \quad (1)$$

where

$$\underline{H} = h\underline{a}^* + k\underline{b}^* + \ell\underline{c}^*, \quad (2)$$

the reciprocal lattice vector,

$$\underline{r}_j = x_j \underline{a} + y_j \underline{b} + z_j \underline{c}, \quad (3)$$

the position vector of the jth atom in the unit cell,

$$\underline{K}_j = m^j_1 \underline{a} + m^j_2 \underline{b} + m^j_3 \underline{c}, \quad (4)$$

a normalized unit vector parallel to the magnetic moment of the jth atom,

$$p_j = (e^2/mc^2)\, \gamma S_j f_j \exp(-B_j \underline{H}\cdot\underline{H}/4), \quad (5)$$

where γ is the magnetic moment of the neutron, S is the spin quantum number of the atom, f is the orbital form factor, B is the isotropic temperature factor and e, m, and c are the usual physical constants,

$$\underline{\varepsilon}\, d\underline{H} = d(h\underline{a}^* + k\underline{b}^* + \ell\underline{c}^*), \quad (6)$$

the unit scattering vector and d is the interplaner spacing.

Let

$$\exp(2\pi i \underline{H}\cdot\underline{r}_j) = A_j + iB_j \qquad (7)$$

where

$$A_j = \cos 2\pi \underline{H}\cdot\underline{r}_j = \cos 2\pi(hx_j + ky_j + \ell z_j)$$

and $$B_j = \sin 2\pi \underline{H}\cdot\underline{r}_j = \sin 2\pi(hx_j + ky_j + \ell z_j).$$

Then

$$F_A(\underline{H}) = \underline{\varepsilon}\,(\underline{\varepsilon}\cdot\Sigma_j p_j \underline{K}_j \cdot A_j) - \Sigma_j p_j \underline{K}_j A_j \qquad (8)$$

$$F_A^2(\underline{H}) = \underline{\varepsilon}\cdot(\underline{\varepsilon}\cdot\Sigma_j p_j \underline{K}_j A_j)\cdot\underline{\varepsilon}(\underline{\varepsilon}\cdot\Sigma_j p_j \underline{K}_j A_j)$$

$$+ (\Sigma_j p_j \underline{K}_j A_j)\cdot(\Sigma_j p_j \underline{K}_j A_j)$$

$$- 2\underline{\varepsilon}(\underline{\varepsilon}\cdot\Sigma_j p_j \underline{K}_j A_j)\cdot(\Sigma_j p_j \underline{K}_j A_j). \qquad (9)$$

In terms of h, k, ℓ, the cell constants a, b, c, α, ƒ, γ and components of the normalized spin vectors, m_1, m_2, and m_3, equation (9) becomes

$$F_A^2(\underline{H}) = -d^2[\Sigma_i p_i A_i (hm_1^i + km_2^i + \ell m_3^i)]^2 +$$

$$\Sigma\Sigma_{ij} p_i A_i p_j A_j \{m_1^i m_1^j a^2 + m_2^i m_2^j b^2 + m_3^i m_3^j c^2 +$$

$$2(m_1^i m_2^j ab\cos\gamma + m_1^i m_3^j ac\cos f + m_2^i m_3^j bc\cos\gamma)\} \qquad (10)$$

A similar expression for $F_B^2(\underline{H})$ is obtained by substituting B for A in equation (10).

Then

$$|F^2(\underline{H})| = F_A^2(\underline{H}) + F_B^2(\underline{H}) \qquad (11)$$

which is the desired result.

Spin vectors read into the machine do not have to be normalized. The direction of the spin vectors is specified by reading in components M_1, M_2, M_3 which define a vector $\underline{M}$ with respect to the unit cell axes. Then for the spin direction of the jth atom,

$$K_j = \left|\frac{\underline{M}_j}{\underline{M}_j}\right| = \frac{M_1\underline{a} + M_2\underline{b} + M_3\underline{c}}{|\underline{M}_j|} \qquad (12)$$

and

$$m_1^j = \left|\frac{M_1}{\underline{M}_j}\right|, \; m_2^j = \left|\frac{M_2}{\underline{M}_j}\right| \text{ and } m_3^j = \left|\frac{M_3}{\underline{M}_j}\right| \qquad (13)$$

where

$$|\underline{M}_j| = (aM_1^j)^2 + (bM_2^j)^2 + (cH_3^j)^2$$

$$+ 2(M_1^jM_2^jab\cos\gamma + M_1^jM_3^jac\cos\beta + M_2^jM_3^jbc\cos\alpha) \qquad (14)$$

Input

Card 1 — Title information, 72 Hollerith characters.

Card 2 — Cell constants a,b,c,α, β,γ, or a,b,c,cosα, cos β, cosγ. Format (6F 10.5)

Card 3 — Number of chemical species (up to 6), number of each species and number of unpaired electrons in the species*. Format (I5, 6(I2, F 8.5))

Cards 4 — One form factor card for each species giving 14 values of f at intervals of sin θ/λ=0.05 starting at sin θ/λ=0. Linear interpolation is made. Format (14F 5.4)

Cards 5 — One card for each atom (maximum of 64) giving x,y,z,B and spin vector components M_1, M_2, M_3 parallel to a,b,c.
(i.e. $\underline{M} = M_1\underline{a} + M_2\underline{b} + M_3\underline{c}$) Format (7F 10.5)

Cards 6 — One card giving h,k,ℓ for each desired plane. Format (3F 5.0)
Program ends when blank card ($\underline{H}$ = 0) is read.
Press start to read complete new set of cards.
If sense switch 2 is on, program reads new data without pause.

*For partially ordered spins, a fractional number of unpaired electrons can be used.

Example

Table 1 lists a sample input and Table 2 the output for the calculation of a few of the magnetic reflection intensities for Fe_2SiO_4. There are two different sets of four magnetic Fe^{+2} ions, each set having a different number of unpaired electrons. Two form factor curves must be read in because the form factor curve is modified according to the number of unpaired electrons before it is used.

Glossary

XNAME	Problem identification
A	a axis
B	b axis
C	c axis
CA	α or cos α
CB	β or cos β
CC	γ or cos γ
SA	sin α
SB	sin β
SC	sin γ
CASTAR	cos α^*
CBSTAR	cos β^*
CCSTAR	cos γ^*
SASTAR	sin α^*
SBSTAR	sin β^*
SCSTAR	sin γ^*
ASTAR	a^*
BSTAR	b^*
CSTAR	c^*
PI	2π
ASQ	a^2
BSQ	b^2
CSQ	c^2
AB	abcos γ
AC	accos β
BC	bccos α

NK	Number of magnetic species
NI(I)	Number of atoms of species I in the cell
SPIN(I)	Number of unpaired electrons in species I.
FØRM(I,J)	The value of the jth point in the form factor of atom of species I. Value stored as f *SPIN(I)*0.539/2.0.
I,J,K	index variables
JMAX	Number of atoms in the unit cell
X(I), Y(I), Z(I)	Atom coordinates, unit cell fractions
T(I)	Temperature factors
X1(I), X2(I), X3(I)	Spin vector components M_1, M_2, M_3
XN	Magnitude of spin vector
Y1(I), Y2(I), Y3(I)	Normalized spin vector components m_1, m_2, m_3
H1, H2, H3	h k ℓ
Q	$1/d^2$
DSQ	d^2
STL	$\sin \theta/\lambda$
SN, NS, TN, SF	Quantities involved in form factor interpolation
E	Form factor for reflection h k ℓ Also used as a temporary variable in the summation of equation (10) in the 300 DO loop.
P	E exp $(-B \sin^2\theta/\lambda^2)$ see equation (5)
ARG	$2\pi (hx_i+ky_i+\ell z_i)$
A1	P cos (ARG)
B1	P sin (ARG)
S1A, S1B	The first summation in equation (10)
S2A, S2B	The second summation in equation (10)
FSQ	$\lvert F(\underline{H})\rvert^2 = F_A^2(\underline{H}) + F_B^2(\underline{H})$
NH1, NH2, NH3	h k ℓ in integer form

IBM

Table 1

FORTRAN CODING FORM

Program: Magnetic Intensities
Programmer:
Date:
Punching Instructions
Graphic
Punch
Card Form # *
Page of
Identification (73–80)

C FOR COMMENT

STATEMENT NUMBER (1–5)	Cont. (6)	FORTRAN STATEMENT (7–72)
	S	FF(2) SI#(4) MAGNETIC INTENSITIES
	4	822 10480 6088
	2	4 4958 4 3962
	1	976 910 814 701 585 476 376 295 227 173 132 103 084
	1	976 910 814 701 585 476 376 295 227 173 132 103 084
	9	871 2973 25 0 0 0 1
	0	129 7207 75 0 0 0 1
	4	871 2207 75 0 0 0 -1
	5	129 7793 25 0 0 0 -1
	0	0 0 0 881 230 143
	0	0 5 0 -881 -230 143
	5	5 0 0 881 -230 -143
	5	5 5 0 -881 230 -143
0		0 1
1		0 0
0		1 0
1		0 1
0		1 1
1		1 0
1		1 1
0		0 0

* A standard card form. IBM electro 888157. is available for punching source statements from this form.

Table 2

FE(2)SIO(4) MAGNETIC INTENSITIES

CELL CONSTANTS

A= 4.822 B= 10.480 C= 6.088

COS ALPHA= 0.00000 COS BETA= 0.00000 COS GAMMA= 0.00000

THERE ARE 2 ATOMS IN THE CELL
4 WITH 4.95800 UNPAIRED ELECTRONS
4 WITH 3.96200 UNPAIRED ELECTRONS

1. .976 .910 .814 .701 .585 .476 .376 .295 .227 .173 .132 .103 .084
1. .976 .910 .814 .701 .585 .476 .376 .295 .227 .173 .132 .103 .084

ATOM	X	Y	Z	B	SPIN VECTORS			NORMALIZED SPIN VECTORS		
1	.987	.279	.250	0.000	0.000	0.000	1.000	0.0000	0.0000	.1642
2	.012	.720	.750	0.000	0.000	0.000	1.000	0.0000	0.0000	.1642
3	.487	.220	.750	0.000	0.000	0.000	-1.000	0.0000	0.0000	-.1642
4	.512	.779	.250	0.000	0.000	0.000	-1.000	0.0000	0.0000	-.1642
5	0.000	0.000	0.000	0.000	.881	.230	1.430	.0882	.0230	.1432
6	0.000	0.000	.500	0.000	-.881	-.230	1.430	-.0882	-.0230	.1432
7	.500	.500	0.000	0.000	.881	-.230	-1.430	.0882	-.0230	-.1432
8	.500	.500	.500	0.000	-.881	.230	-1.430	-.0882	.0230	-.1432

H	K	L	F SQD
0	0	1	2.8794
1	0	0	66.8008
0	1	0	7.2014
1	0	1	.7648
0	1	1	10.4761
1	1	0	.1410
1	1	1	1.1321

```
C      MAGNETIC INTENSITIES
C      OUTPUT WILL BE PRINTED WITH SENSE SWITCH 1 OFF AND PUNCHED
C      WITH SENSE SWITCH 1 ON
C      IF SENSE SWITCH 2 IS ON PROGRAM WILL READ NEW DATA SET
C      WHEN FINISHED. IF SENSE SWITCH 2 IS OFF PROGRAM STOPS.
C      PRESS START TO READ NEW DATA SET.
       DIMENSIONXNAME(18),NI(6),FORM(6,14),SPIN(6),X(64),Y(64),Z(64),
      1T(64),X1(64),X2(64),X3(64),Y1(64),Y2(64),Y3(64),P(64),A1(64),
      2B1(64)
     1 FORMAT (18A4)
     2 FORMAT (1H1,18A4)
     3 FORMAT (6F10.5)
     4 FORMAT (I5,6(I2,F8.5))
     5 FORMAT (14F5.4)
     6 FORMAT (7F10.5)
     7 FORMAT (3I4,F10.4)
     8 FORMAT (25H    H    K    L F SQUARED)
     9 FORMAT (/72H ATOM   X      Y      Z      B      SPIN VECTORS      NORMA
      1LIZED SPIN VECTORS)
    10 FORMAT (I4,7F6.3,3X,3F7.4)
    11 FORMAT (/10X,14HCELL CONSTANTS/
      110H         A=F8.3,11H          B=F8.3,12H          C=F8.3/
      210HCOS ALPHA=F8.5,11H  COS BETA=F8.5,12H  COS GAMMA=F8.5/)
    12 FORMAT (3F5.0)
    13 FORMAT (4F10.5,I5,2F10.5)
    14 FORMAT (9F7.3)
    15 FORMAT (/21H   H   K   L    F SQD)
    16 FORMAT (I3, 5H WITH,F8.5,19H UNPAIRED ELECTRONS)
    17 FORMAT ( 9HTHERE ARE,I3,18H ATOMS IN THE CELL)
    18 FORMAT(F3.0,13F5.3)
C      READ PROBLEM IDENTIFICATION
C*     READ INPUT TAPE 10,1,XNAME
    99 READ 1,XNAME
C*     WRITE OUTPUT TAPE 9,2,XNAME
       IF(SENSE SWITCH 1)1000,1001
 1000  PUNCH 1,XNAME
       GO TO 1002
 1001  PRINT 1,XNAME
 1002  CONTINUE
C      READ CELL CONSTANTS
C      CELL ANGLES IN DEGREES OR AS COSINES
C*     READ INPUT TAPE 10,3,A,B,C,CA,CB,CC
       READ 3,A,B,C,CA,CB,CC
       IF(CA-1.0)101,101,100
  100  CA=COSF(CA/57.29578)
  101  IF(CB-1.0)103,103,102
  102  CB=COSF(CB/57.29578)
  103  IF(CC-1.0)105,105,104
  104  CC=COSF(CC/57.29578)
  105  SA=SQRTF(1.0-CA*CA)
       SB=SQRTF(1.0-CB*CB)
       SC=SQRTF(1.0-CC*CC)
       CASTAR=(CB*CC-CA)/(SB*SC)
       CBSTAR=(CA*CC-CB)/(SA*SC)
       CCSTAR=(CA*CB-CC)/(SA*SB)
```

```
      SASTAR=SQRTF(1.0-CASTAR*CASTAR)
      SBSTAR=SQRTF(1.0-CBSTAR*CBSTAR)
      SCSTAR=SQRTF(1.0-CCSTAR*CCSTAR)
      ASTAR=1.0/(A*SB*SCSTAR)
      BSTAR=1.0/(B*SA*SCSTAR)
      CSTAR=1.0/(C*SA*SBSTAR)
C*    WRITE OUTPUT TAPE 9,11,A,B,C,CA,CB,CC
      IF(SENSE SWITCH 1)1003,1004
 1003 PUNCH 11,A,B,C,CA,CB,CC
      GO TO 1005
 1004 PRINT 11,A,B,C,CA,CB,CC
 1005 CONTINUE
      PI=2.0*3.1415926
      ASQ=A*A
      BSQ=B*B
      CSQ=C*C
      AB=A*B*CC
      AC=A*C*CB
      BC=B*C*CA
C     READ NUMBER OF SPECIES, NUMBER OF EACH SPECIES AND
C       NUMBER OF UNPAIRED ELECTRONS IN EACH SPECIES
C     MAXIMUM NUMBER OF SPECIES=6
C*    READ INPUT TAPE 10,4,NK,(NI(I),SPIN(I),I=1,NK)
      READ 4,NK,(NI(I),SPIN(I),I=1,6)
C*    WRITE OUTPUT TAPE 9,17,NK
      IF(SENSE SWITCH 1)1060,1061
 1061 PRINT 17,NK
      GO TO 1062
 1060 PUNCH 17,NK
 1062 DO 1065 I=1,NK
      IF(SENSE SWITCH 1)1063,1064
 1063 PUNCH 16,NI(I),SPIN(I)
      GO TO 1065
 1064 PRINT 16,NI(I),SPIN(I)
C*    WRITE OUTPUT TAPE 9,16,NI(I),SPIN(I)
 1065 CONTINUE
      DO 110 I=1,NK
C     READ MAGNETIC FORM FACTOR TABLE, ONE FOR EACH SPECIES. EACH TABLE
C       HAS 14 VALUES AT INTERVALS OF 0.05 IN SIN(THETA)/LAMBDA
C       STARTING AT SIN(THETA)/LAMBDA=0.0  LINEAR INTERPOLATION IS MADE.
C*    READ INPUT TAPE 10,5,(FORM(I,J),J=1,14)
      READ 5,(FORM(I,J),J=1,14)
      IF(SENSE SWITCH 1)1070,1071
 1070 PUNCH 18, (FORM(I,J),J=1,14)
      GO TO 1072
 1071 PRINT 18, (FORM(I,J),J=1,14)
 1072 CONTINUE
      DO 109 J=1,14
  109 FORM(I,J)=FORM(I,J)*SPIN(I)/2.0*0.539
  110 CONTINUE
      J=0
      DO 120 I=1,NK
      K=NI(I)
      DO 115 M=1,K
      J=J+1
```

```
C      READ X,Y,Z,TEMPERATURE FACTOR, AND THREE COMPONENTS OF THE SPIN
C        VECTOR IN REAL CRYSTAL SPACE FOR EACH ATOM IN THE UNIT CELL.
C*     READ INPUT TAPE 10,6,X(J),Y(J),Z(J),T(J),X1(J),X2(J),X3(J)
  115 READ 6,X(J),Y(J),Z(J),T(J),X1(J),X2(J),X3(J)
  120 CONTINUE
C NORMALIZE SPIN VECTORS
      JMAX=J
      DO 130 I=1,JMAX
      XN=(A*X1(I))*(A*X1(I))+(B*X2(I))*(B*X2(I))+(C*X3(I))*(C*X3(I))+2.0
     1*((A*X1(I))*(B*X2(I))*CC+(A*X1(I))*(C*X3(I))*CB+(B*X2(I))*
     2(C*X3(I))*CA)
      XN=SQRTF(XN)
      Y1(I)=X1(I)/XN
      Y2(I)=X2(I)/XN
  130 Y3(I)=X3(I)/XN
C*131 WRITE OUTPUT TAPE 9,9
      IF(SENSE SWITCH 1)1006,131
 1006 PUNCH 9
      GO TO 1007
  131 PRINT 9
 1007 CONTINUE
      DO 135 I=1,JMAX
C*135 WRITE OUTPUT TAPE 9,10,I,X(I),Y(I),Z(I),T(I),X1(I),X2(I),X3(I),
C*   1Y1(I),Y2(I),Y3(I)
      IF(SENSE SWITCH 1) 1008,1009
 1008 PUNCH 10,I,X(I),Y(I),Z(I),T(I),X1(I),X2(I),X3(I),Y1(I),Y2(I),Y3(I)
      GO TO 135
 1009 PRINT 10,I,X(I),Y(I),Z(I),T(I),X1(I),X2(I),X3(I),Y1(I),Y2(I),Y3(I)
  135 CONTINUE
C*     WRITE OUTPUT TAPE 9,15
      IF(SENSE SWITCH 1)1011,1010
 1010 PRINT 15
      GO TO 1012
 1011 PUNCH 15
 1012 CONTINUE
C      READ HKL FOR THE DESIRED REFLECTIONS. PROBLEM ENDS WHEN BLANK
C        CARD IS READ.
C*200 READ INPUT TAPE 10,12,H1,H2,H3
  200 READ 12,H1,H2,H3
      IF(ABSF(H1)+ABSF(H2)+ABSF(H3))201,500,201
  201 Q=(H1*ASTAR)*(H1*ASTAR)+(H2*BSTAR)*(H2*BSTAR)+(H3*CSTAR)*(H3*CSTAR
     1)+2.0*((H2*BSTAR)*(H3*CSTAR)*CASTAR+(H1*ASTAR)*(H3*CSTAR)*CBSTAR+
     2(H1*ASTAR)*(H2*BSTAR)*CCSTAR)
      DSQ=1.0/Q
C     FORM FACTOR INTERPOLATION
      STL=0.5/SQRTF(DSQ)
      SN=STL/0.05
      NS=SN
      TN=NS
      SF=SN-TN
      J=0
      DO 210 I=1,NK
      K=NI(I)
      E=FORM(I,NS+1)-(FORM(I,NS+1)-FORM(I,NS+2))*SF
  204 DO 205 M=1,K
```

```
      J=J+1
  205 P(J)=E*EXPF(-0.25*Q*T(J))
  210 CONTINUE
      DO 230 I=1,JMAX
      ARG=PI*(H1*X(I)+H2*Y(I)+H3*Z(I))
      A1(I)=COSF(ARG)*P(I)
  230 B1(I)=SINF(ARG)*P(I)
      S1A=0.0
      S2A=0.0
      S1B=0.0
      S2B=0.0
C     COMPUTE MAGNETIC SCATTERING
      DO 300 I=1,JMAX
      E=            (Y1(I)*H1+Y2(I)*H2+Y3(I)*H3)
      S1A=S1A+A1(I)*E
      S1B=S1B+B1(I)*E
      DO 250 K=1,JMAX
      E=Y1(I)*Y1(K)*ASQ+Y2(I)*Y2(K)*BSQ+Y3(I)*Y3(K)*CSQ+
     1(Y1(I)*Y2(K)*AB+Y1(I)*Y3(K)*AC+Y2(I)*Y3(K)*BC)*2.0
      S2A=S2A+E*A1(I)*A1(K)
  250 S2B=S2B+E*B1(I)*B1(K)
  300 CONTINUE
      FSQ=-DSQ*(S1A*S1A+S1B*S1B)+S2A+S2B
      NH1=H1
      NH2=H2
      NH3=H3
C*    WRITE OUTPUT TAPE 9,7,NH1,NH2,NH3,FSQ
      IF(SENSE SWITCH 1)1015,1016
 1015 PUNCH 7,NH1,NH2,NH3,FSQ
      GO TO 1017
 1016 PRINT 7,NH1,NH2,NH3,FSQ
 1017 CONTINUE
      GO TO 200
  500 IF(SENSE SWITCH 2)99,501
  501 PAUSE
      GO TO 99
      END
```